Demo 19 May 2008

Initialization

```
Clear["Global`*"]
```

Basic

Calculator

```
3 + 5 (* <enter> or <shift-return> but not <return> *)
```

```
5 / 3
```

```
5.0 / 3
```

```
N[5 / 3]
```

```
5 / 3 // N
```

```
3 * 5
```

```
3 × 5
```

Algebra

```
Solve[x^2 + α x == β, x]
```

Calculus

```
∂_x x^3
```

$$\int x^2 \, dx$$

$$\int_0^\infty e^{-x} \, dx$$

$$\int_0^\infty e^{-a x} \, dx$$

```
Integrate[Exp[-a x], {x, 0, Infinity}, Assumptions → {a > 0}]
```

Functions

```
f[x_] := x Sin[1/x] (* Note square brackets & capital letter on Sin *)
```

```
f[a]

f[☺]

f[3]

f[3] // N

f'[x]
```

Graphing

```
Plot[Sin[x] + Sin[1.6 x], {x, 0, 12 π}]

Options[Plot]

Plot[
 Sin[x] + Sin[1.6 x],
 {x, 0, 12 π},
 Filling → Axis,
 Frame → True,
 FrameLabel → {x, y},
 RotateLabel → False
]

Manipulate[
 Plot[
  Sin[x] + Sin[α x],
  {x, 0, 12 π},
  Filling → Axis,
  Frame → True,
  FrameLabel → {x, y},
  RotateLabel → False
 ],
 {α, -9, 9}
]

Manipulate[
 Plot[
  Sin[x] + Sin[α x], {x, 0, 12 π},
  PlotRange → {-2, 2},
  Filling → Axis,
  Frame → True,
  FrameLabel → {x, y},
  RotateLabel → False,
  PlotPoints → 100
 ],
 {{α, -20, "ratio"}, -9, 9}
]

Plot3D[Sin[x y], {x, -π, π}, {y, -π, π}]

Manipulate[
 Plot3D[
  Sin[x y],
  {x, -π, π}, {y, -π, π},
  Mesh → None,
  ColorFunction → "TemperatureMap",
  PlotStyle → Opacity[a]
 ],
 {{a, 1}, 0, 1}
]
```

Advanced

```
Prime[10^9]
```

```
N[π, 10]
```

```
RealDigits[N[π, 10]]
```

```
RealDigits[N[π, 10]][[1]]
```

```
Partition[RealDigits[N[π, 9]][[1]], 3]
```

```
n = 1 000 000;
Graphics[
 Raster[
  Partition[RealDigits[N[π, n]][[1]], √n],
  Automatic,
  {0, 9}
 ]
]
```

```
n = 1 000 000;
Style[
 Graphics[
  Raster[
   Partition[RealDigits[N[π, n]][[1]], √n],
   Automatic,
   {0, 9}
  ],
  ImageSize → √n, PlotRangePadding → None
 ],
 Magnification → 1, Antialiasing → False
]
```

ODEs

Damped SHO

Equations

```
equation = {m x''[t] == -γ x'[t] - k x[t]};
```

```
start = {x[0] == 1, x'[0] == 0};
```

Exact Solution

```
solution = DSolve[equation, x[t], t]

solution = DSolve[equation ⋃ start, x[t], t]
```

Numerical Solution

```
parameters = {k → 1, m → 1, γ → 0.1};

tMin = 0;
tMax = 30;

solution = NDSolve[equation ⋃ start /. parameters, x, {t, tMin, tMax}]

xS[t_] = x[t] /. Flatten[solution]

Plot[{xS[t], xS'[t]}, {t, tMin, tMax}]

ParametricPlot[{xS[t], xS'[t]}, {t, tMin, tMax}]
```

Kepler

Equations

```
rVec[t_] = {x[t], y[t], z[t]};

rMag[t_] = √(rVec[t].rVec[t]) ;

rHat[t_] = rVec[t] / rMag[t];

equation = Thread[rVec''[t] == -GM / rMag[t]^2 rHat[t]]

start = {Thread[rVec[0] == {1, 0, 0}], Thread[rVec'[0] == {0, 0.9, 0}]}
```

Numerical Solution

```
parameters = {GM → 1};

tMin = 0;
tMax = 5;

solution = NDSolve[equation ⋃ start /. parameters, rVec[t], {t, tMin, tMax}]

ParametricPlot3D[rVec[t] /. solution, {t, tMin, tMax}]

rVecS[t_] = rVec[t] /. Flatten[solution];

Graphics3D[
 {
  Red, Opacity[.3], Polygon[{{2, 2, 0}, {-2, 2, 0}, {-2, -2, 0}, {2, -2, 0}}],
  Yellow, Sphere[{0, 0, 0}, 0.3],
  Blue, Sphere[rVecS[0], 0.2]
 }
]
```

```
Manipulate[
 Graphics3D[
  {
   Red, Opacity[.3], Polygon[{{2, 2, 0}, {-2, 2, 0}, {-2, -2, 0}, {2, -2, 0}}],
   Yellow, Opacity[.7], Sphere[{0, 0, 0}, 0.3],
   Blue, Opacity[.7], Sphere[rVecS[t], 0.2]
  },
  Boxed → False
 ],
 {t, tMin, tMax}
]
```